<p>I have a regional cluster for redundancy. In this cluster I want to create a node-pool in just 1 zone in this region. Is this configuration possible? reason I trying this is, I want to run service like RabbitMQ in just 1 zone to avoid split, and my application services running on all zones in the region for redundancy.</p>
<p>I am using terraform to create the cluster and node pools, below is my config for creating region cluster and zone node pool</p>
<pre><code>resource &quot;google_container_cluster&quot; &quot;regional_cluster&quot; {
  provider       = google-beta
  project        = &quot;my-project&quot;
  name           = &quot;my-cluster&quot;
  location       = &quot;us-central1&quot;
  node_locations = [&quot;us-central1-a&quot;, &quot;us-central1-b&quot;, &quot;us-central1-c&quot;]

  master_auth {
    username = &quot;&quot;
    password = &quot;&quot;

    client_certificate_config {
      issue_client_certificate = false
    }
  }
}

resource &quot;google_container_node_pool&quot; &quot;one_zone&quot; {
  project            = google_container_cluster.regional_cluster.project
  name               = &quot;zone-pool&quot;
  location           = &quot;us-central1-b&quot;
  cluster            = google_container_cluster.regional_cluster.name

  node_config {
    machine_type    = var.machine_type
    image_type      = var.image_type
    disk_size_gb    = 100
    disk_type       = &quot;pd-standard&quot;
  }
}

</code></pre>
<p>This throws an error message</p>
<pre><code>error creating NodePool: googleapi: Error 404: Not found: projects/my-project/zones/us-central1-b/clusters/my-cluster., notFound

</code></pre>

RandomQuests
<p>Found out that <code>location</code> in <code>google_container_node_pool</code> should specify cluster master's region/zone. To actually specify the node-pool location <code>node_locations</code> should be used. Below is the config that worked</p>
<pre><code>resource &quot;google_container_cluster&quot; &quot;regional_cluster&quot; {
  provider       = google-beta
  project        = &quot;my-project&quot;
  name           = &quot;my-cluster&quot;
  location       = &quot;us-central1&quot;
  node_locations = [&quot;us-central1-a&quot;, &quot;us-central1-b&quot;, &quot;us-central1-c&quot;]

  master_auth {
    username = &quot;&quot;
    password = &quot;&quot;

    client_certificate_config {
      issue_client_certificate = false
    }
  }
}

resource &quot;google_container_node_pool&quot; &quot;one_zone&quot; {
  project            = google_container_cluster.regional_cluster.project
  name               = &quot;zone-pool&quot;
  location           = google_container_cluster.regional_cluster.location
  node_locations     = [&quot;us-central1-b&quot;]
  cluster            = google_container_cluster.regional_cluster.name

  node_config {
    machine_type    = var.machine_type
    image_type      = var.image_type
    disk_size_gb    = 100
    disk_type       = &quot;pd-standard&quot;
  }
}
</code></pre>
